



# Protocol Audit Report

Version 1.0

*io10*

January 5, 2025

# Protocol Audit Report

io10

04/01/2025

Prepared by: [io10] Lead Auditors:

- io10

## Table of Contents

- Table of Contents
- T-Swap Protocol Summary
- Disclaimer
- Risk Classification
- Audit Details
- Executive Summary
  - Issues found
- Findings
- High
- Low

## T-Swap Protocol Summary

This project is meant to be a permissionless way for users to swap assets between each other at a fair price. You can think of T-Swap as a decentralized asset/token exchange (DEX). T-Swap is known as an Automated Market Maker (AMM) because it doesn't use a normal "order book" style exchange, instead it uses "Pools" of an asset. It is similar to Uniswap. To understand Uniswap, please watch this video: [Uniswap Explained](#)

## Disclaimer

io10 makes all effort to find as many vulnerabilities in the code in the given time period, but holds no responsibilities for the findings provided in this document. A security audit by me is not an endorsement of the underlying business or product. The audit was time-boxed and the review of the code was solely on the security aspects of the Solidity implementation of the contracts.

## Risk Classification

		Impact		
		High	Medium	Low
Likelihood	High	H	H/M	M
	Medium	H/M	M	M/L
	Low	M	M/L	L

We use the CodeHawks severity matrix to determine severity. See the documentation for more details.

## Audit Details

- Commit Hash: e643a8d4c2c802490976b538dd009b351b1c8dda
- In Scope:

```
1 ./src/  
2 #-- PoolFactory.sol  
3 #-- TSwapPool.sol
```

- Solc Version: 0.8.20
- Chain(s) to deploy contract to: Ethereum
- Tokens:
  - Any ERC20 token

Actors / Roles

- **Liquidity Providers:** Users who have liquidity deposited into the pools. Their shares are represented by the LP ERC20 tokens. They gain a 0.3% fee every time a swap is made.
- **Users:** Users who want to swap tokens.

## Executive Summary

The TSwap protocol audit revealed critical vulnerabilities that compromise its core functionality and user trust. Most notably, issues with protocol invariants, fee calculations, and missing safeguards could lead to system instability, overcharging of users, and susceptibility to manipulation.

The audit highlighted that certain design choices, such as added user incentives, break the protocol's fundamental mathematical assumptions, potentially destabilizing the liquidity pool. Additionally, inconsistencies in fee structures and a lack of slippage protection expose users to unexpected costs and unfair trading conditions. Missing deadline enforcement in key functions also risks denial-of-service scenarios and user frustration.

Addressing these issues is paramount to maintaining the protocol's integrity and ensuring predictable, fair outcomes for users. Key recommendations include revising core logic to uphold protocol invariants, ensuring consistency in fee calculations, and implementing safeguards against slippage and missed deadlines.

By resolving these vulnerabilities, TSwap can strengthen its foundation, safeguard user trust, and continue fostering a reliable decentralized trading experience.

## Issues found

Severity	Number Of Issues Found
High	5
Medium	0
Low	1
Informational	0
Total	6

## Findings

### High

[H-1] Protocol Core Invariant Broken which breaks down the core tswap system

**Description:** The core invariant of the protocol is detailed in the documentation of the tswap protocol and it is broken due to added functionality to reward users for using tswap. See below:

Code

```
1  /**
2      * @notice Swaps a given amount of input for a given amount of
3      *         output tokens.
4      * @dev Every 10 swaps, we give the caller an extra token as an
5      *         extra incentive to keep trading on T-Swap.
6      * @param inputToken ERC20 token to pull from caller
7      * @param inputAmount Amount of tokens to pull from caller
8      * @param outputToken ERC20 token to send to caller
9      * @param outputAmount Amount of tokens to send to caller
10     */
11     function _swap(
12         IERC20 inputToken,
13         uint256 inputAmount,
14         IERC20 outputToken,
15         uint256 outputAmount
16     ) private {
17         if (
18             _isUnknown(inputToken) ||
19             _isUnknown(outputToken) ||
20             inputToken == outputToken
21         ) {
22             revert TSwapPool__InvalidToken();
23         }
24
25         swap_count++;
26         if (swap_count >= SWAP_COUNT_MAX) {
27             swap_count = 0;
28             outputToken.safeTransfer(msg.sender, 1
29                                     _000_000_000_000_000_000);
30         }
31         emit Swap(
32             msg.sender,
33             inputToken,
34             inputAmount,
35             outputToken,
36             outputAmount
37         );
38     }
```

```
36         inputToken.safeTransferFrom(msg.sender, address(this),
37             inputAmount);
38         outputToken.safeTransfer(msg.sender, outputAmount);
39     }
```

in the natspec, you will see that it says Every 10 swaps, we give the caller an extra token as an extra incentive to keep trading on T-Swap. This is what breaks the assertion. In the code, you can see a `swap_count` variable being incremented and once it goes past a certain threshold, an extra amount is sent to `msg.sender`. This token incentive is what breaks the invariant of this protocol.

**Impact:** The tswap system fails due to the protocol invariant breaking

**Proof of Concept:** See the `Tswapinvarianttests.t.sol` file for the invariant tests and `Handler.sol` file for the invariant test that exposes the assertion break. Run `forge test --mt invariant_coreinvariantTswap -vvvv` to see results. If fuzz results are hard to read, here is the unit test version that mimics the fuzzer calls and the invariant break.

Code

```
1 function test_10swapsbreaksinvariant() public {
2     vm.startPrank(liquidityProvider);
3     weth.approve(address(pool), 100e18);
4     poolToken.approve(address(pool), 100e18);
5     pool.deposit(100e18, 100e18, 100e18, uint64(block.timestamp));
6     vm.stopPrank();
7
8     vm.startPrank(user);
9     uint256 expected = 9e18;
10    poolToken.approve(address(pool), 300e18);
11    for (uint256 i = 0; i < 9; i++) {
12        pool.swapExactInput(
13            poolToken,
14            5e18,
15            weth,
16            1e18,
17            uint64(block.timestamp)
18        );
19    }
20    uint256 preoutputreserves = weth.balanceOf(address(pool));
21    uint256 expectedoutputAmount = pool.getOutputAmountBasedOnInput(
22        20e18,
23        poolToken.balanceOf(address(pool)),
24        weth.balanceOf(address(pool))
25    );
26    pool.swapExactInput(
27        poolToken,
28        20e18,
29        weth,
30        7e18,
```

```
31         uint64(block.timestamp)
32     );
33     uint256 postoutputreserves = weth.balanceOf(address(pool));
34     uint256 actualoutputAmount = preoutputreserves -
        postoutputreserves;
35     console.log(expectedoutputAmount, actualoutputAmount);
36     assert(actualoutputAmount > expectedoutputAmount);
37     vm.stopPrank();
38 }
```

In the above test, a user performs 10 swaps. Before the 10th swap, the expected output amount is measured as well as the weth balance of the pool. After the 10th swap, the weth balance of the pool is measured and these values are compared to the expected amount of weth to leave the pool and you will see that the actualoutputamount is 1 weth more than expected which accounts for the 1eth incentive given to a user after every 10 swaps which breaks the protocol invariant.

**Recommended Mitigation:** Remove token incentive from `TSwapPool::_swap` function to maintain protocol core invariant.

[H-2] No deadline checks in the `TSwapPool::deposit` function which causes a DOS for the user calling deposit function who sets a deadline

**Description:** See deposit function:

Code

```
1 // @notice Adds liquidity to the pool
2     /// @dev The invariant of this function is that the ratio of WETH,
3     PoolTokens, and LiquidityTokens is the same
4     /// before and after the transaction
5     /// @param wethToDeposit Amount of WETH the user is going to
6     deposit
7     /// @param minimumLiquidityTokensToMint We derive the amount of
8     liquidity tokens to mint from the amount of WETH the
9     user is going to deposit, but set a minimum so they know approx
10    what they will accept //q if the minimum deviates a lot from
11    the actual amount of liquidity tokens minted, how does this give
12    the user any idea what they will accept ?
13
14    /// @param maximumPoolTokensToDeposit The maximum amount of pool
15    tokens the user is willing to deposit, again it's
16    derived from the amount of WETH the user is going to deposit //
17    q how is this derived if the user has to enter a random amount ?
18    /// @param deadline The deadline for the transaction to be
19    completed by
20    function deposit(
21        uint256 wethToDeposit,
22        uint256 minimumLiquidityTokensToMint,
23        uint256 maximumPoolTokensToDeposit,
```

```
15         uint64 deadline //bug no deadline checks here with
           revertIfDeadlinePassed(deadline)
16     )
17     external
18     revertIfZero(wethToDeposit)
19     returns (uint256 liquidityTokensToMint)
20 {
21     if (wethToDeposit < MINIMUM_WETH_LIQUIDITY) {
22         revert TSwapPool__WethDepositAmountTooLow(
23             MINIMUM_WETH_LIQUIDITY,
24             wethToDeposit
25         );
26     }
27     if (totalLiquidityTokenSupply() > 0) {
28         uint256 wethReserves = i_wethToken.balanceOf(address(this))
29         ;
30         uint256 poolTokenReserves = i_poolToken.balanceOf(address(
31             this));
32         uint256 poolTokensToDeposit =
33             getPoolTokensToDepositBasedOnWeth(
34                 wethToDeposit
35             );
36         if (maximumPoolTokensToDeposit < poolTokensToDeposit) {
37             revert TSwapPool__MaxPoolTokenDepositTooHigh(
38                 maximumPoolTokensToDeposit,
39                 poolTokensToDeposit
40             );
41         }
42         // We do the same thing for liquidity tokens. Similar math.
43         liquidityTokensToMint =
44             (wethToDeposit * totalLiquidityTokenSupply()) /
45             wethReserves;
46         if (liquidityTokensToMint < minimumLiquidityTokensToMint) {
47             revert TSwapPool__MinLiquidityTokensToMintTooLow(
48                 minimumLiquidityTokensToMint,
49                 liquidityTokensToMint
50             );
51         }
52         _addLiquidityMintAndTransfer(
53             wethToDeposit,
54             poolTokensToDeposit,
55             liquidityTokensToMint
56         );
57     } else {
58         // This will be the "initial" funding of the protocol. We
59         // are starting from blank here!
60         // We just have them send the tokens in, and we mint
61         // liquidity tokens based on the weth
62         _addLiquidityMintAndTransfer(
63             wethToDeposit,
```



```
60         maximumPoolTokensToDeposit,  
61         wethToDeposit  
62     );  
63     liquidityTokensToMint = wethToDeposit;  
64 }  
65 }
```

There are no deadline checks in the deposit function which means that if a user sets a deadline for a deposit and expects it to fail past a certain block, the deposit will be processed leading to a DOS.

**Impact:** Unexpected behaviour from user calling deadline function

**Proof of Concept:**

Code

```
1  function test_depositwdeadline() public {  
2      vm.startPrank(liquidityProvider);  
3      weth.approve(address(pool), 100e18);  
4      poolToken.approve(address(pool), 100e18);  
5      vm.warp(5000);  
6      vm.roll(5);  
7      pool.deposit(100e18, 100e18, 100e18, uint64(block.timestamp -  
8          300));  
9      assertEq(pool.balanceOf(liquidityProvider), 100e18);  
10     assertEq(weth.balanceOf(liquidityProvider), 100e18);  
11     assertEq(poolToken.balanceOf(liquidityProvider), 100e18);  
12  
13     assertEq(weth.balanceOf(address(pool)), 100e18);  
14     assertEq(poolToken.balanceOf(address(pool)), 100e18);  
15 }
```

**Recommended Mitigation:** Include the `TSwapPool:: revertIfDeadlinePassed(deadline)` in the deposit function which includes deadline logic.

[H-3] Fees are inconsistent between `TSwapPool:: swapExactInput` and `TSwapPool:: swapExactOutput` leading to swappers paying more in fees when using `TSwapPool:: swapExactOutput`

**Description:** The logic for `TSwapPool:: swapExactInput` was based on the following formula which baked the 0.3% fee into each swap:

```
1  uint256 inputAmountMinusFee = inputAmount * 997;  
2      uint256 numerator = inputAmountMinusFee * outputReserves;  
3      uint256 denominator = (inputReserves * 1000) +  
4          inputAmountMinusFee;  
5      return numerator / denominator;
```

For `TSwapPool:: swapExactOutput`, the logic was:

```
1 (inputReserves * outputAmount * 10000)((outputReserves - outputAmount)
   * 997);
```

As you can see, the fee calculation is skewed due to the extra 0 in the above formula.

**Impact:** As a result of the above description, users who choose to swap using `TSwapPool::swapExactOutput` end up paying a 3% fee instead of the aforementioned 0.3% intended by the protocol.

### Proof of Concept:

Code

```
1  function testCollectFeesInput() public {
2      vm.startPrank(liquidityProvider);
3      weth.approve(address(pool), 100e18);
4      poolToken.approve(address(pool), 100e18);
5      pool.deposit(100e18, 100e18, 100e18, uint64(block.timestamp));
6      vm.stopPrank();
7
8      vm.startPrank(user);
9      uint256 expected = 9e18;
10     poolToken.approve(address(pool), 10e18);
11     pool.swapExactInput(
12         poolToken,
13         10e18,
14         weth,
15         expected,
16         uint64(block.timestamp)
17     );
18     vm.stopPrank();
19
20     vm.startPrank(liquidityProvider);
21     pool.approve(address(pool), 100e18);
22     pool.withdraw(100e18, 90e18, 100e18, uint64(block.timestamp));
23     console.log(
24         weth.balanceOf(liquidityProvider) +
25         poolToken.balanceOf(liquidityProvider)
26     );
27     assertEq(pool.totalSupply(), 0);
28     assert(
29         weth.balanceOf(liquidityProvider) +
30         poolToken.balanceOf(liquidityProvider) >
31         200e18
32     );
33 }
34
35 function testCollectFeesOutput() public {
36     vm.startPrank(liquidityProvider);
37     weth.approve(address(pool), 100e18);
```

```
38     poolToken.approve(address(pool), 100e18);
39     pool.deposit(100e18, 100e18, 100e18, uint64(block.timestamp));
40     vm.stopPrank();
41
42     vm.startPrank(user);
43     uint256 expected = 9e18;
44     poolToken.approve(address(pool), 10e18);
45     weth.approve(address(pool), 100e18);
46     pool.swapExactOutput(
47         weth,
48         poolToken,
49         expected,
50         uint64(block.timestamp)
51     );
52     vm.stopPrank();
53
54     vm.startPrank(liquidityProvider);
55     pool.approve(address(pool), 100e18);
56     pool.withdraw(100e18, 90e18, 90e18, uint64(block.timestamp));
57     console.log(
58         weth.balanceOf(liquidityProvider) +
59         poolToken.balanceOf(liquidityProvider)
60     );
61     assertEq(pool.totalSupply(), 0);
62     assert(
63         weth.balanceOf(liquidityProvider) +
64         poolToken.balanceOf(liquidityProvider) >
65         200e18
66     );
67 }
```

Inspecting the logs from both of these tests will highlight the fee discrepancies when calling `TSwapPool:: swapExactInput` vs `TSwapPool:: swapExactOutput`

**Recommended Mitigation:** Refactor the `TSwap::getInputAmountBasedOnOutput` function which is used in the `TSwapPool:: swapExactOutput` to calculate input amount. See below:

```
1 - ((inputReserves * outputAmount) * 10000)
2 + ((inputReserves * outputAmount) * 1000)
3 ((outputReserves - outputAmount) * 997);
```

[H-4] Lack of slippage protection in `TSwapPool:: swapExactOutput` which allows for MEV and causes users to pay unexpected input amounts for swaps

**Description:** In the `TSwapPool:: swapExactOutput` function, notice how there is no `maxInputAmount` parameter. This is a problem because if a big transaction frontruns the users transaction, it will cause a slip in the `inputAmount` calculation which can lead to the user paying more than they wanted to in order to swap their tokens which can be an issue if the slippage is high.

**Impact:** User paying unexpected and excessive inputAmounts for swaps

**Proof of Concept:**

Code

```
1  function test_unexpectedslippage() public {
2      vm.startPrank(liquidityProvider);
3      weth.approve(address(pool), 100e18);
4      poolToken.approve(address(pool), 100e18);
5      pool.deposit(100e18, 100e18, 100e18, uint64(block.timestamp));
6      vm.stopPrank();
7
8      uint256 expected = 9e18;
9      uint256 inputreserves = weth.balanceOf(address(pool));
10     uint256 outputreserves = poolToken.balanceOf(address(pool));
11     uint256 expecteduserinput = pool.getInputAmountBasedOnOutput(
12         expected,
13         inputreserves,
14         outputreserves
15     );
16     console.log(expecteduserinput);
17
18     vm.startPrank(user2);
19     weth.approve(address(pool), 100e18);
20     pool.swapExactOutput(
21         weth,
22         poolToken,
23         expected,
24         uint64(block.timestamp)
25     );
26     vm.stopPrank();
27
28     uint256 newinputreserves = weth.balanceOf(address(pool));
29     uint256 newoutputreserves = poolToken.balanceOf(address(pool));
30     uint256 actualuserinput = pool.getInputAmountBasedOnOutput(
31         expected,
32         newinputreserves,
33         newoutputreserves
34     );
35
36     vm.startPrank(user);
37     weth.approve(address(pool), 300e18);
38     pool.swapExactOutput(
39         weth,
40         poolToken,
41         expected,
42         uint64(block.timestamp)
43     );
44     vm.stopPrank();
45
46     console.log(actualuserinput, expecteduserinput);
```

```

47         assert(actualuserinput > expecteduserinput);
48     }

```

**Recommended Mitigation:** Include a `maxInputAmount` parameter in `TSwapPool::swapExactOutput` function that will allow the user to set a maximum input amount they are willing to pay for a specified output and if `inputAmount` goes above this, then tx should revert. See modified function below:

Code

```

1  function swapExactOutput(
2      IERC20 inputToken,
3      IERC20 outputToken,
4  +   uint256 maxInputAmount,
5      uint256 outputAmount,
6      uint64 deadline
7  )
8
9      public
10     revertIfZero(outputAmount)
11     revertIfDeadlinePassed(deadline)
12     returns (uint256 inputAmount)
13 {
14     uint256 inputReserves = inputToken.balanceOf(address(this));
15     uint256 outputReserves = outputToken.balanceOf(address(this));
16
17     inputAmount = getInputAmountBasedOnOutput(
18         outputAmount,
19         inputReserves,
20         outputReserves
21     );
22
23 +     if (inputAmount > maxInputAmount) {
24 +         revert TSwapPool__InputTooHigh(inputAmount, maxInputAmount
25 +     );
26     }
27
28     _swap(inputToken, inputAmount, outputToken, outputAmount);
29 }

```

[H-5] Wrong Logic in `TSwapPool::sellPoolTokens` which forces users to sell more pool tokens than their intended amount

**Description:** In `TSwapPool::sellPoolTokens`, see the description of the function below:

Code

```

1
2  /**

```

```
3      * @notice wrapper function to facilitate users selling pool tokens
      *       in exchange of WETH
4      * @param poolTokenAmount amount of pool tokens to sell
5      * @return wethAmount amount of WETH received by caller
6      */
7      function sellPoolTokens(
8          uint256 poolTokenAmount
9      ) external returns (uint256 wethAmount) {
10         return
11             swapExactOutput(
12                 i_poolToken,
13                 i_wethToken,
14                 poolTokenAmount,
15                 uint64(block.timestamp)
16             );
17     }
```

`TswapPool::poolTokenAmount` is a parameter where the user should enter the amount of pool tokens they want to sell. By calling `TswapPool::swapExactOutput` in this function, `TswapPool::poolTokenAmount` is set as the output amount. i.e. the amount of weth that the user wants to receive. As a result, the function ends up taking a larger input amount than `TswapPool::poolTokenAmount` which is not expected behaviour and in some cases, can lead to users swapping a larger amount of pool tokens than they intended

**Impact:** Users are forced to sell more pool tokens than their intended amount

**Proof of Concept:**

Code

```
1
2  function test_sellpooltokenswronglogic() public {
3      vm.startPrank(liquidityProvider);
4      weth.approve(address(pool), 100e18);
5      poolToken.approve(address(pool), 100e18);
6      pool.deposit(100e18, 100e18, 100e18, uint64(block.timestamp));
7      vm.stopPrank();
8
9      vm.startPrank(user);
10     uint256 expected = 10e18;
11     poolToken.approve(address(pool), 200e18);
12     uint256 inputAmount = pool.getInputAmountBasedOnOutput(
13         expected,
14         poolToken.balanceOf(address(pool)),
15         weth.balanceOf(address(pool))
16     );
17     pool.sellPoolTokens(expected);
18     assert(inputAmount > expected);
19     vm.stopPrank();
20 }
```

**Recommended Mitigation:** Replace `TswapPool::swapExactOutput` function call in `TswapPool::sellPoolTokens` with `TswapPool::swapExactInput`. See below:

Code

```
1
2 /**
3     * @notice wrapper function to facilitate users selling pool tokens
4     *         in exchange of WETH
5     * @param poolTokenAmount amount of pool tokens to sell
6     * @return wethAmount amount of WETH received by caller
7     */
8     function sellPoolTokens(
9 +         uint256 poolTokenAmount,
10 +         uint256 minOutputAmount,
11 ) external returns (uint256 wethAmount) {
12     return
13 -         swapExactOutput(
14 -             i_poolToken,
15 -             i_wethToken,
16 -             poolTokenAmount,
17 -             uint64(block.timestamp)
18 -         );
19 +         swapExactInput(
20 +             i_poolToken,
21 +             poolTokenAmount,
22 +             i_wethToken,
23 +             minOutputAmount,
24 +             uint64(block.timestamp)
25 +         )
26 }
```

## Low

[L-1] Event Details not emitted correctly which leads to potential migration issues and misleading information to users

**Description:** The event is initialised as:

```
1 event LiquidityAdded(
2     address indexed liquidityProvider,
3     uint256 wethDeposited,
4     uint256 poolTokensDeposited
5 );
```

and emitted as:

```
1 emit LiquidityAdded(msg.sender, poolTokensToDeposit, wethToDeposit);
```

**Impact:** Potential Migration Issues and emitted event can mislead users on tokens they have added to the pool

**Proof of Concept:** See description

**Recommended Mitigation:** Refactor emitted event to:

```
1 emit LiquidityAdded(msg.sender, wethToDeposit, poolTokensToDeposit);
```